
9 References

References make it easy to pass data to and from subroutines and let the subroutines modify your variables, arrays, and hashes.  If you know some Java or C++, you'll find Perl references similar to references in Java and C++.  If you know C, you'll see that references are like pointers.  And if you don't know what I'm talking about, you soon will!

9.1 What is a reference?

A variable holds some data in your computer's memory.  This data usually takes one of the forms we've seen already: a number, a string, an array, or a hash.

#

# prints out 'Just some string.'

#

my $str = "Just some string.";

print $str;

A reference or reference variable is like a scalar variable.  It holds just one piece of data like a scalar variable.  But the type of data is contains is what sets it apart from regular variable.  A reference contains an address in the computer's memory.  Usually, this address will be the address of some data of interest in the computer's memory.

First let's talk about how you actually find the address of data in the computer.  Getting the address is as simple as adding a back slash to a variable.

#

# Creating a string and assigning it to a

# variable tells Perl to store the string

# in memory.

#

$str = "Just some string.";

print $str;

print \$str;

This will print out some thing similar to

Just some string.

SCALAR(0x8102c64)

In the second print statement, \$str, prints out the address, or memory location of the data that $str contains.  The first part of this output, SCALAR, tells us that the address is pointing to SCALAR data.  If we print out the address of an array or hash, it will print out ARRAY or HASH.  The second half of the print out, 0x8102c64, is a hexadecimal number that represents a location in the computer's memory, the location containing the first byte of the string "Just some string".

To create a reference, just assign the address to a scalar variable.

#

# Create a scalar variable and assign a string to it.

#

my $str = "Just some string.";

#

# Get the address of $str and assign the address to a reference.

#

my $ref = \$str;

#

# This will print out:

#    Just some string.

#    SCALAR(0x8102c64)

#    SCALAR(0x8102c64)

#

print "$str\n";

print \$str,"\n";

print $ref;

References have some special properties that you'll lean below.  In order for a reference to be created correctly, it has to be assigned like I just did above.  Don't assign a reference using and address directly like this

my $ref = "0x8102c64";

This would not be a reference, it would just be some variable with a hexadecimal number.

9.2 Dereferencing a reference

Ok, so now you know how to make a reference to a scalar.  But once you have this reference, how do you get at the data it points to?  To get at the data, you need to dereference the reference.  When the reference points to a scalar, you dereference it by prepending the reference with a $ (so that there are two $'s).

#

# This prints out:

#   Just some string.

#   Just some string.

#

my $str = "Just some string.\n";

my $ref = \$str;

print $str;

print $$ref;

9.3 Changing a value with a reference

Not only can look at a value by dereferencing a reference, you can also change the value.

#

# Prints out:

#   Just some string.

#   A new string.

#

my $str = "Just some string.\n";

my $ref = \$str;

print $str;

$$ref = "A new string.\n";

print $str;

9.4 Scalar references and subroutines

When you pass a variable to a subroutine, it's passed by value.  What this means is that when you put a variable into the argument list of a subroutine, a copy of that variable's data is sent to the subroutine.  This means that if the subroutine changes the data it was given, it will be unchanged in the code that called the subroutine.

#!/usr/bin/perl -w

use strict;

#

my $i = 1;

print "\$i in main before demo is $i.\n";

&demo( $i );

print "\$i in main after demo is $i.\n";

exit;

#

sub demo {


my $i = shift;


$i = 33;


return;

}

This will print out

$i in main before demo is 1.

$i in main before demo is 1.

There is nothing unusual here because demo()'s copy of $i is local since we declared it with my.

But what if we wanted demo to be able to change the value of main's $i?  We know of two ways so far.  First we could leave out my and use globals.  However, we learned that we should avoid using global variables whenever possible.  Second we could have demo() return a value and assign that return value to main's $i.

#!/usr/bin/perl -w

use strict;

#

my $i = 1;

print "\$i in main before demo is $i.\n";

$i = &demo( $i );

print "\$i in main after demo is $i.\n";

exit;

#

sub demo {


my $i = shift;


$i = 33;


return $i;

}

This prints out

$i in main before demo is 1.

$i in main after demo is 33.

The third way is to use references.

#!/usr/bin/perl -w

use strict;

#

my $i = 1;

print "\$i in main before demo is $i.\n";

&demo( \$i );

print "\$i in main after demo is $i.\n";

exit;

#

sub demo {


my $i_ref = shift;


$$i_ref = 33;


return;
}

This will print out

$i in main before demo is 1.

$i in main after demo is 33.
If you look at the above main code, there is nothing different except for the call to demo().

&demo( \$i );

In the previous versions of main, demo()'s argument was $i and a copy of $i's contents is sent to the subroutine.  In this case, demo()'s argument is \$i and a copy the address to $i's data is sent to demo().

In the actual subroutine, the copy of the address is received by $i_ref (a reference to $i).  In the subroutine, we change the data the reference points to by dereferencing the reference and assigning a new value

$$i_ref = 33;

9.5 Array references

Creating a reference to an array just as easy as it was with scalars

my @array = ("one", "two", "three");

my $arrayRef =  \@array;

You can dereference an array reference in several ways.  The first is to prepend the array reference with an @ to get the whole array.  You would use this method if you wanted to copy an entire array with an array reference or you are an array reference in a function that expects an array.

# This prints:

#   one two three

#   one two three

my @a = ("one", "two", "three");

my $ar = \@a;

print "@a\n";

print "@$ar\n";

# This prints:

#   one

#   two

#   three

foreach my $num ( @$ar ) { print "$num\n"; }

You can add an index to get at individual array elements by putting braces around the array reference, putting a $ in front, and following it with the index of the array element you want.  Study the following example to see how it's done.

# This prints:

#   one

#   two

#   three

for( my $i=0; $i<@$ar; $i++ ) {


print "${$ar}[$i]\n";

}

The alternate method of dereferencing array references is by means of the infix dereference operator, ->, also more commonly known as the arrow operator.  Using the same  @a and $ar above

# This prints:

#   one two three

print "$ar->[0] $ar->[1] $ar->[2]";

Using the arrow operator is seen more often because it is used in many languages, such as C, to access data in a data structure, such as an array, with a pointer.  Additionally, the arrow operator is used extensively in object-oriented Perl to access object methods and data.

9.6 Hash references

Using references with hashes is almost identical to the array methods.

my %hash = ("one" => 1, "two" => 2, "three" => 3);

my $hashRef = \%hash;

foreach my $key ( sort keys %$hashRef ) {


#


# print the key


#


print "KEY: $key\n";


#


# One way to dereference the hash reference


#


print "VALUE: ${$hashRef}{$key}\n";


#


# Another way to dereference the hash reference


# with an arrow operator


#


print "VALUE: $hashRef->{$key}\n";

}

9.7 Array references and subroutines

Using array and hash references with subroutines is just as simple as with scalars.  Infact, if you want to pass more than one array and/or hash to a subroutine, you are required to use references.  To see why references are required, let's see what happens when we use pass-by-value instead.

#!/usr/bin/perl -w

#

my @a = (1, 2, 3);

my @b = (4, 5, 6);

demo( @a, @b );

exit;

sub demo {


print "@_ \n";

}

This code will print out

1 2 3 4 5 6

Now look carefully at demo()'s arguments.  We are giving it two arrays, @a and @b.  Inside demo(), all of the contents of @a and @b are copied to one array, @_.  Since all of the elements of @a and @b are copied to this one array, we can't tell which elements of @_ came from @a or @b.

If you want to keep the elements of @a and @b separate when they are passed to a subroutine, use array references as the subroutine's arguments.

#!/usr/bin/perl -w

#

my @a = (1, 2, 3);

my @b = (4, 5, 6);

demo( \@a, \@b );

exit;

sub demo {


my $ar1 = shift;  # ar means array reference


my $ar2 = shift;


print "First array contains @$ar1\n";


print "Second array contains @$ar2\n";

}

This code will print out

First array contains 1 2 3

Second array contains 4 5 6

You can handle hash references the same way.

#!/usr/bin/perl -w

#

my %a = ("one"=>1, "two"=>2, "three"=>3);

my %b = ("four"=>4, "five"=>5, "six"=>6);

demo( \%a, \%b );

exit;

sub demo {


my $hr1 = shift;  # hr means hash reference


my $hr2 = shift;


...


some code to manipulate the hashes


...

}

And of course you can mix data types in your subroutine arguments.

#!/usr/bin/perl -w

#

my %a = ("one"=>1, "two"=>2, "three"=>3);

my @b = (4, 5, 6);

my $c = "VII";

demo( \%a, \@b, \$c );

exit;

sub demo {


my $hr1 = shift;  # hr means hash reference


my $ar2 = shift;  # ar means array reference


my $sr = shift;   # sr means scalar reference


...


some code to manipulate the referenced data


...

}

9.8 Exercises

1.  Write a subroutine that takes two arrays as arguments and swaps each value between them.  In other words, if @a = (1,2,3) and @b = (4,5,6), then after passing these arrays to the subroutine they would be @a = (4,5,6) and @b = (1,2,3).  Write this subroutine so that is doesn't use references.

2. Repeat exercise 1, however, this time you must use references.

3. Write a subroutine called initHash() that takes one hash reference and two array references as arguments.  The input hash should be empty at the beginning of the subroutine.  The hash will be populated with key-value pairs in the subroutine.  One of the array contains a list of keys; these keys will become the keys of the empty hash.  The other array should contain a list of values that will become the values of the hash.  The code that calls this subroutine might look like this

my @keys = (... you put in a list of keys here ...);

my @vals = (... you put a list of values here ...);

my %hash = ();

initHash( \%hash, \@keys, \@vals );

